Solving equations

Exercise 1.1

```
> restart;
> f := (x) -> 2*x^4-2222*x^3+224220*x^2-2222000*x+2000000;
```

$$f := x \mapsto 2x^4 - 2222x^3 + 224220x^2 - 2222000x + 2000000 \quad \textbf{(1)}$$

```
> solve(f(x)=0,{x});
```

$$\{x=1\}, \{x=10\}, \{x=1000\}, \{x=100\} \quad \textbf{(2)}$$

Whenever a polynomial $f(x)$ has a root $x=a$, the term $x-a$ is a factor of $f(x)$. This indicates that $f(x)$ should be a multiple of the polynomial $g(x) = (x-1)\,(x-10)\,(x-100)\,(x-1000)$. However, the coefficient of x^4 in $f(x)$ is 2, whereas the coefficient of x^4 in the expansion of $g(x)$ is 1. Thus, the multiplier must be 2, and we must have

$$f(x) = 2\,g(x) = 2\,(x-1)\,(x-10)\,(x-100)\,(x-1000)$$

Maple confirms this as follows:

```
> factor(f(x));
```

$$2\,(x-1)\,(x-10)\,(x-1000)\,(x-100) \quad \textbf{(3)}$$

Exercise 1.2

```
> restart;
> y := x^4-x^3-x^2-x/8+1/64;
```

$$y := x^4 - x^3 - x^2 - \frac{1}{8}x + \frac{1}{64} \quad \textbf{(4)}$$

```
> _EnvExplicit := true:
> solve(y=0,x);
```

$$\frac{1}{4} + \frac{\sqrt{6}}{4} + \frac{\sqrt{3}}{4} + \frac{\sqrt{2}}{4}, \frac{1}{4} + \frac{\sqrt{6}}{4} - \frac{\sqrt{3}}{4} - \frac{\sqrt{2}}{4}, \frac{1}{4} - \frac{\sqrt{6}}{4} + \frac{\sqrt{3}}{4} - \frac{\sqrt{2}}{4}, \frac{1}{4} - \frac{\sqrt{6}}{4} - \frac{\sqrt{3}}{4} + \frac{\sqrt{2}}{4} \quad \textbf{(5)}$$

```
> sols := solve(y=0,{x});
```

$$sols := \left\{x = \frac{1}{4} + \frac{\sqrt{6}}{4} + \frac{\sqrt{3}}{4} + \frac{\sqrt{2}}{4}\right\}, \left\{x = \frac{1}{4} + \frac{\sqrt{6}}{4} - \frac{\sqrt{3}}{4} - \frac{\sqrt{2}}{4}\right\}, \left\{x = \frac{1}{4} - \frac{\sqrt{6}}{4} + \frac{\sqrt{3}}{4} - \frac{\sqrt{2}}{4}\right\}, \left\{x = \frac{1}{4} - \frac{\sqrt{6}}{4} - \frac{\sqrt{3}}{4} + \frac{\sqrt{2}}{4}\right\} \quad \textbf{(6)}$$

In each of the four solutions, the sign attached to $\sqrt{6}$ is the product of the signs attached to $\sqrt{2}$ and $\sqrt{3}$. Using this observation, we see that the four solutions are $(1\pm\sqrt{2})(1\pm\sqrt{3})/4$.

```
> z := 16*x^3-24*x^2-6*x+2;
```

$$z := 16\,x^3 - 24\,x^2 - 6\,x + 2 \tag{7}$$

```
> simplify(subs(sols[1],z),symbolic);
```

$$-\sqrt{2} \tag{8}$$

```
> simplify(subs(sols[2],z),symbolic);
```

$$\sqrt{2} \tag{9}$$

```
> simplify(subs(sols[3],z),symbolic);
```

$$\sqrt{2} \tag{10}$$

```
> simplify(subs(sols[4],z),symbolic);
```

$$-\sqrt{2} \tag{11}$$

```
> seq(simplify(subs(sols[i],z),symbolic),i=1..4);
```

$$-\sqrt{2}, \sqrt{2}, \sqrt{2}, -\sqrt{2} \tag{12}$$

Exercise 1.3

```
> restart;
> _EnvExplicit := true;
```

$$\_EnvExplicit := true \tag{13}$$

```
> y := x^3-3*x+1;
```

$$y := x^3 - 3\,x + 1 \tag{14}$$

```
> solve(y=0,{x});
```

$$\left\{x = \frac{\left(-4 + 4\,\mathrm{I}\sqrt{3}\right)^{1/3}}{2} + \frac{2}{\left(-4 + 4\,\mathrm{I}\sqrt{3}\right)^{1/3}}\right\}, \left\{x = -\frac{\left(-4 + 4\,\mathrm{I}\sqrt{3}\right)^{1/3}}{4} - \frac{1}{\left(-4 + 4\,\mathrm{I}\sqrt{3}\right)^{1/3}} + \frac{\mathrm{I}\sqrt{3}\left(\frac{\left(-4 + 4\,\mathrm{I}\sqrt{3}\right)^{1/3}}{2} - \frac{2}{\left(-4 + 4\,\mathrm{I}\sqrt{3}\right)^{1/3}}\right)}{2}\right\}, \left\{x = -\frac{\left(-4 + 4\,\mathrm{I}\sqrt{3}\right)^{1/3}}{4} - \frac{1}{\left(-4 + 4\,\mathrm{I}\sqrt{3}\right)^{1/3}} \tag{15}$$

$$-\frac{I\sqrt{3}\left(\frac{\left(-4+4\,I\sqrt{3}\right)^{1/3}}{2}-\frac{2}{\left(-4+4\,I\sqrt{3}\right)^{1/3}}\right)}{2}\Bigg\}$$

```
> sols := fsolve(y=0,{x});
```

$$sols := \{x=-1.879385242\},\ \{x=0.3472963553\},\ \{x=1.532088886\} \qquad \textbf{(16)}$$

```
> plot(y,x=-2..2);
```



The only negative root is $x=-1.879$, which is **sols[1]**. Maple notation for the gradient $\frac{dy}{dx}$ is **diff(y,x)**. We can thus find the gradient at the negative root as follows:

```
> subs(sols[1],diff(y,x));
```

$$7.59626666 \qquad \textbf{(17)}$$

Exercise 1.4

```
> g := (x) -> (b^2 - c^2 + (1 + c^2)*x)/(1-c^2 + c^2*x);
```

$$g := x \mapsto \frac{b^2-c^2+\left(c^2+1\right)x}{1-c^2+c^2x} \qquad \textbf{(18)}$$

(a)

```
> sols := solve(g(x)=x,{x});
```

$$sols := \left\{x=-\frac{b-c}{c}\right\},\ \left\{x=\frac{b+c}{c}\right\} \qquad \textbf{(19)}$$

(b)

The above solutions involve division by c, so they do not make sense when $c=0$. We must therefore check separately what happens when $c=0$. In that case, $g(x)$ simplifies as follows:

```
> subs(c=0,g(x));
```

$$b^2+x \qquad \textbf{(20)}$$

The equation $g(x) = x$ thus becomes $x + b^2 = x$. This is always true if $b = 0$, and never true if $b \neq 0$.

(c) Now return to the case $c \neq 0$. The two solutions can then be written as $x = 1 - \frac{b}{c}$ and $x = 1 + \frac{b}{c}$. These are the same (and both equal to 1) if $b = 0$, but are different otherwise.

(d) Our conclusion is as follows:

- If $b = 0 = c$ then $g(x) = x$ for all x.
- If $b = 0 \neq c$ then the only solution to $g(x) = x$ is $x = 1$.
- If $b \neq 0 = c$ then there are no solutions.
- If $b \neq 0 \neq c$ then there are precisely two different solutions, namely $1 + \frac{b}{c}$ and $1 - \frac{b}{c}$.

Exercise 2.1

```
> f := (x) -> sin(Pi*x+exp(-x));
```

$$f := x \mapsto \sin\left(\pi x + e^{-x}\right) \qquad (21)$$

```
> solve(f(x)=0,{x});
```

$$\left\{x = \mathrm{LambertW}\left(-\frac{1}{\pi}\right)\right\}, \left\{x = \mathrm{LambertW}\left(-1, -\frac{1}{\pi}\right)\right\} \qquad (22)$$

(a)

```
> plot(f(x),x=0..10);
```



The graph looks rather like a sine wave, with roots at $x = 1,2,3,4$ and so on. This makes sense, because as soon as x becomes reasonably large, the term e^{-x} becomes very small, so $f(x)$ is close to $\sin(\mathrm{Pi}\, x)$. The positive roots of $\sin(\mathrm{Pi}\, x)$ are exactly $x = 1,2,3,4$ and so on, so we expect the roots of $f(x)$ to be approximately the same.

(b) We now find some roots more precisely:

```
> fsolve(f(x)=0,x);
```
$$-0.5538270366 \quad (23)$$

```
> fsolve(f(x)=0,x=2);
```
$$1.954935731 \quad (24)$$

```
> seq(fsolve(f(x)=0,x=n),n=1..10);
```
0.8661259484, 1.954935731, 2.983894990, 3.994135661, 4.997850630, 5.999210365, 6.999709655, 7.999893208, 8.999960716, 9.999985549 **(25)**

(c)

```
> r := (n) -> n-exp(-n)/Pi-exp(-2*n)/Pi^2;
```
$$r := n \mapsto n - \frac{e^{-n}}{\pi} - \frac{e^{-2n}}{\pi^2} \quad (26)$$

```
> seq(evalf(r(n)),n=1..10);
```
0.8691880059, 1.955065679, 2.983901134, 3.994135962, 4.997850645, 5.999210366, 6.999709655, 7.999893208, 8.999960715, 9.999985549 **(27)**

This is very close to our answer in (b). We can calculate the differences as follows:

```
> seq(fsolve(f(x)=0,x=n)-evalf(r(n)),n=1..10);
```
$-0.0030620575, -0.000129948, -6.144\ 10^{-6}, -3.01\ 10^{-7}, -1.5\ 10^{-8}, -1.\ 10^{-9}, 0., 0., 1.\ 10^{-9}, 0.$ **(28)**

Exercise 3.1

(a)

```
> plot(tan(Pi*x)^2-3,x=-3..3,-3..3,numpoints=200);
```



(b)

The roots are at $-\frac{8}{3}, -\frac{7}{3}, -\frac{5}{3}, -\frac{4}{3}, -\frac{2}{3}, -\frac{1}{3}, \frac{1}{3}, \frac{2}{3}, \frac{4}{3}, \frac{5}{3}, \frac{7}{3}, \frac{8}{3}$ and so on.

In other words, for every integer n, there is a root at $n - \frac{1}{3}$, and another one at $n + \frac{1}{3}$.

(c)

```
> solve(tan(Pi*x)^2=3,{x});
```

$$\left\{x = \frac{1}{3}\right\}, \left\{x = -\frac{1}{3}\right\} \qquad (29)$$

(d)

```
> _EnvAllSolutions := true;
```

$$\_EnvAllSolutions := true \qquad (30)$$

```
> solve(tan(Pi*x)^2=3,{x});
```

$$\left\{x = \frac{1}{3} + \_Z1\sim\right\}, \left\{x = -\frac{1}{3} + \_Z2\sim\right\} \qquad (31)$$

Exercise 4.1

```
> restart;
> eqns := {
   x/2 + y/3 + z/4 = 1,
   x/3 + y/4 + z/5 = 2,
   x/4 + y/5 + z/6 = 3
  };
```

$$eqns := \left\{\frac{x}{2} + \frac{y}{3} + \frac{z}{4} = 1, \frac{x}{3} + \frac{y}{4} + \frac{z}{5} = 2, \frac{x}{4} + \frac{y}{5} + \frac{z}{6} = 3\right\} \qquad (32)$$

```
> sol := solve(eqns,{x,y,z});
```

$$sol := \{x = 132, y = -600, z = 540\} \qquad (33)$$

```
> sol := solve({x/2+y/3+z/4=1,x/3+y/4+z/5=2,x/4+y/5+z/6=3},{x,y,z})
  ;
```

$$sol := \{x = 132, y = -600, z = 540\} \qquad (34)$$

```
> subs(sol,x^2+y^2+z^2);
```

$$x^2 + (x^3 - 3x + 1)^2 + z^2 \qquad (35)$$

Exercise 4.2

(a)

```
> solve({p+q+r=0,p+2*q+3*r=1});
```

$$\{p = -1 + r, q = 1 - 2r, r = r\} \qquad (36)$$

Note the equation $p = p$ in the solution, indicating that p can take any value. This means that there are infinitely many different solutions.

(b)

```
> solve({u+v=1001,u+2*v=1002,u+3*v=1006});
> [solve({u+v=1001,u+2*v=1002,u+3*v=1006})];
```

$$[\ ] \tag{37}$$

Maple gives an empty response, indicating that there are no solutions. This is easy to see directly: if we subtract the first two equations we get $v=1$, whereas subtracting the second and third equations gives $v=4$, showing that the equations are not consistent.

(c)

```
> solve({
    x +   y +   z = 2,
    x + 2*y + 3*z = 2,
    x + 4*y + 9*z = 2
   },
   {x,y,z}
  );
```

$$\{x=2, y=0, z=0\} \tag{38}$$

This has a unique, fully-determined solution.